Projekt Milchkaffee

Nina Verspohl, Josef Riese, 08.01.2003.

```
> restart:
```

Die allgemeine DGL lautet:

```
> ode:= D(u)(t) = - beta *(u(t) - A);
```

$$ode := \mathrm{D}(u)(t) = -\beta\,(\mathrm{u}(t) - A)$$

allgemein: T ist die Anfangstemperatur zum Zeitpunkt 0 (Anfangsbedingung)

```
> ic:=u(0)=T;
```

$$ic := \mathrm{u}(0) = T$$

allgemeine Lösung der DGL

```
> lsg:=dsolve( { ode,ic}, u(t) );
```

$$lsg := \mathrm{u}(t) = A + \mathbf{e}^{(-\beta\,t)}\,(-A + T)$$

verschiedene Schreibweisen: einmal, wie wir es gewohnt sind diese zu schreiben, ein anderes mal mit 'Maple- Schreibweise'

```
> convert( ode, diff );
  convert( %, D );
```

$$\frac{d}{dt}\,\mathrm{u}(t) = -\beta\,(\mathrm{u}(t) - A)$$

$$\mathrm{D}(u)(t) = -\beta\,(\mathrm{u}(t) - A)$$

Einsetzten der Lösung

```
> subs( lsg, convert( ode, diff ) );
```

$$\frac{\partial}{\partial t}\,(A + \mathbf{e}^{(-\beta\,t)}\,(-A + T)) = -\beta\,\mathbf{e}^{(-\beta\,t)}\,(-A + T)$$

Kontrolle, ob die Lösung richtig ist, also ob beide Seiten der Gleichung identisch sind

```
> eval( % );
```

$$-\beta\,\mathbf{e}^{(-\beta\,t)}\,(-A + T) = -\beta\,\mathbf{e}^{(-\beta\,t)}\,(-A + T)$$

nun die Rechnung mit konkreten Werten

```
> ick := u(0)= 80;  #Temperatur des Kaffees zu Beginn
  A:= 20; #Außentemperatur von 20˚C (Raumtemperatur)
```

$$ick := \mathrm{u}(0) = 80$$

$$A := 20$$

Lösung für die "Kaffeefunktion"

```
> lsgk:=unapply(rhs(dsolve( { ode, ick })),t);
```

$$lsgk := t \rightarrow 20 + 60\,\mathbf{e}^{(-\beta\,t)}$$

gesucht ist nun unser Beta, um die Anfangsbedingung zu erfüllen:

```
> solve (lsgk(600)-42,beta);
```

$$-\frac{1}{600}\,\ln\left(\frac{11}{30}\right)$$

Beta als Fließkommazahl:

```
> beta := evalf(%);
```

$$\beta := 0.001672170182$$

Zur Verschönerung nocheinmal die "Kaffeefunktion" mit dem errechneten beta Wert

```
> lsgk:=unapply(rhs(dsolve( { ode, ick })),t);
```

$$lsgk := t \rightarrow 20 + 60\, \mathbf{e}^{\left(-\frac{836085091}{500000000000}t\right)}$$

Anfangstemperatur der Milch:

```
> icm := u(0)= 8;
```

Lösung für die "Milchfunktion"

```
> lsgm:=unapply(rhs(dsolve( { ode, icm })),t);
```

$$icm := \mathrm{u}(0) = 8$$

$$lsgm := t \rightarrow 20 - 12\, \mathbf{e}^{\left(-\frac{836085091}{500000000000}t\right)}$$

Nach 10 Minuten (=600 Sekunden) hat die Milch die Temperatur

```
evalf(lsgm(600));
```

$$15.60000000$$

Die Mischung hat also nach 10 Minuten (=600 Sek) eine Temperatur von

```
> evalf((lsgm(600)+lsgk(600))/2);
```

Somit stimmt sowohl Raumtemperatur, als auch der Parameter Beta unseres Anfangswertproblems

$$28.80000000$$

Wann müssen nun Milch und Kaffee zusammen geschüttet werden, um eine Temperatur von 42 ˚C erreichen ???

```
z:=solve((lsgk(t)+lsgm(t))/2=42,t);
```

$$z := -\frac{500000000000}{836085091} \ln\left(\frac{11}{12}\right)$$

```
> evalf (z);
```

$$52.03500091$$

Nach 52,04 Sekunden ist die gewünschte Bedingung erfüllt, dass 42˚C Trinktemperatur erreicht sin
Probe:

```
evalf(lsgm(52.04));evalf(lsgk(52.04)); (%+%%)/2;
```

$$9.00009195$$

$$74.99954024$$

$$41.99981610$$

ii)

```
> with(plots):
```

```
Warning, the name changecoords has been redefined
```



```
Warning, the name changecoords has been redefined
```


Die Darstellung zeigt die Temperaturkurven in Oskars Fall: Bestehend aus Milch (gelb), Kaffe (braun), Mischung (ocka) und Raum(weiß)

```
> K:=plot(lsgk(t), t=0..z,0..100,color=[brown]):
```

```
M:=plot(lsgm(t), t=0..z,0..100,color=[orange]):
macro(milchkaffee = COLOR(RGB, 1, 0.8, 0.4)):
MI:=plot((lsgm(t)+lsgk(t))/2,
t=z..300,0..100,color=milchkaffee):
R:=plot(20, t=0..300,color=wheat): #Raumtemperatur von 20˚C
Z:=plot(42, t=0..300,color=blue): #Zieltemperatur von 42˚C
display({K,M,MI,R,Z},title=‘Oskars Milchkaffee‘);
```



iii)
Olga trinkt ihren Kaffee nur mit 10% Milch, aber sie mischt Kaffee und Milch sofort und lässt die Mischung abkühlen:
 Die entsprechenden DGL in ihrem Fall lauten nun:

Anfangsbedingung mit einem Teil 8˚C kalter Milch und neun Teilen 80˚C warmen Kaffee

```
> icol := u(0)= (1*8+9*80)/10;
```

$$icol := \mathrm{u}(0) = \frac{364}{5}$$

Lösung der entsprechenden DGL

```
> lsgol:=unapply(rhs(dsolve( { ode, icol })),t);
```

$$lsgol := t \rightarrow 20 + \frac{264}{5}\,\mathbf{e}^{\left(-\frac{836085091}{500000000000}\,t\right)}$$

Die Trinktemperatur soll wie oben 42˚C betragen

```
> OL:=evalf(solve(lsgol(t)=42,t));
```

$$OL := 523.5524151$$

Hier nun die Temperaturkurve für Olgas Methode:

```
> macro(milchkaffeeol = COLOR(RGB, 0.7, 0.4, 0.2)):
  plot(lsgol(t),t=0..600,0..100,color=milchkaffeeol,title=‘Olgas
  Milchkaffee‘);
```



iv)
Man könnte vermuten, dass es sinnvoller ist, erst die Milch hinzuzufügen und dann die Mischung abkühlen zu lassen, da das ganze dann auf einem niedrigerem Niveau beginnt abzukühlen. Ob dies stimmt, wollen wir nun untersuchen.

v)
Jetzt Olgas Mischungsverhältnis mit Oskars Methode, also erst unmittelbar vorm Trinken mischen:

```
> OS:=solve((9*lsgk(t)+1*lsgm(t))/10=42,t);
```

Offenbar besteht kein Unterschied zwischen beiden Methoden?!?

$$OS := -\frac{500000000000}{836085091} \ln\left(\frac{5}{12}\right)$$

```
> evalf (OS);
```

$$523.5524151$$

Diese Darstellung zeigt die Temperaturkurven beider Verfahren im Vergleich: Bestehend aus Milch (gelb), Oskars Kaffee (dunkelbraun), Oskars Mischung (ocka), Olgas Kaffeemischung (mittelbraun) und Raum(weiß).

```
> K2:=plot(lsgk(t), t=0..OS,0..100,color=[brown]):
  M2:=plot(lsgm(t), t=0..OS,0..100,color=[orange]):
  macro(milchkaffee = COLOR(RGB, 1, 0.8, 0.4)):
  MI2:=plot((lsgm(t)+9*lsgk(t))/10,
  t=OS..600,0..100,color=milchkaffee):
  R:=plot(20, t=0..600,color=wheat): #Raumtemperatur von 20˚C
  macro(milchkaffeeol = COLOR(RGB, 0.7, 0.4, 0.2)):
  OL2:=plot(lsgol(t),t=0..OL,0..100,color=milchkaffeeol):
  display({K2,M2,MI2,R,OL2},title=‘Vergleich der Verfahren
  (9:1)‘);
```




```
> display({K2,M2,MI2,R,OL2},view=[500...540,40..50],axes=boxed,tit
  le='Vergleich der Verfahren im Ausschnitt');
```




```
> OL3:=plot(lsgol(t),t=500..600,0..100,color=milchkaffeeol):
> display(
  {OL3,display({K2,M2,MI2})},insequence=true,view=[500...540,10..5
  0],title=`"Film":Vergleich der Verfahren im Ausschnitt`);
```



Nun überlegen wir, was passiert, wenn sich der Raum miterwärmt:
Da sowohl Olgas als auch Oskars Methode zu gleichem Ergebnis führen, betrachten wir im Folgenden nur Olga. Wir nehmen an, dass das Problem Raum-Milchkaffe analog zum Problem Mich-Kaffee ist. Wir vermuten also einen Grenzwert, an welchen sich beide Kurven annähern, wenn sich Raum und Kaffee angeglichen haben.
Angenommen: Im ausgeglichenen Zustand 22˚C

```
> A:=22:
  icol := u(0)= (1*8+9*80)/10;
```

$$icol := \mathrm{u}(0) = \frac{364}{5}$$

Lösung der sich nun ergebenden DGL:

```
> lsgol:=unapply(rhs(dsolve( { ode, icol })),t);
```

$$lsgol := t \rightarrow \frac{18393872000}{836085091} + \frac{212365613124}{4180425455}\,\mathbf{e}^{\left(-\frac{836085091}{500000000000}\,t\right)}$$

Die Zeit für die gewünschteTrinktemperatur ist somit:

```
> evalf(solve(lsgol(t)=42,t));
```

$$557.4576625$$

Wie man sich auch vorher schon denken konnte, dauert das Abkühlen nun etwas länger, da ja der Raum wärmer geworden ist.

Variationen

i)

Für Oskar bleibt der Trinktemperaturzeitpunkt natürlich bei etwa 52 Sek., der Kaffee ist somit lang
Zeit vorm Verlassen des Gebäudes trinkfertig.

Wenn Olga nach drei Minuten das Gebäude verlässt, hat die Kaffeemischung folgende Temperatur:

```
> A:= 10;
```

$$A := 10$$

```
> icolga:= u(180)= evalf (lsgol (180));
```

$$icolga := u(180) = 59.59630302$$

```
> lsgolga:= unapply(rhs(dsolve( { ode, icolga }, u(t) )),t):
> zeitolga:= evalf(solve(lsgolga(t)=42,t));
```

$$zeitolga := 442.0429409$$

ii)

Wir vermuten, dass wieder beide Strategien identisch sind (s.o.).

```
> icm2:= u(180) = evalf (lsgm (180));
```

$$icm2 := u(180) = 11.11898354$$

```
> ick2:= u(180) = evalf (lsgk (180));
```

$$ick2 := u(180) = 64.40508231$$

Daraus ergeben sich dann neue DGL'en:

```
> lsgm2:= unapply(rhs(dsolve( { ode, icm2 }, u(t) )),t):
> lsgk2:= unapply(rhs(dsolve( { ode, ick2 }, u(t) )),t):
```

Olgas Mischungsverhältnis mit Oskars Methode:

```
> Osk:=solve((9*lsgk2(t)+1*lsgm2(t))/10=42,t);
```

$$Osk := -\frac{500000000000}{836085091} \ln\left(\frac{32000000000}{49076472433}\right) + 180$$

```
> evalf(Osk);
```

$$435.7418170$$

Da ist also ein Unterschied zu erkennen! Oskars Methode ist schneller!

```
> K4:=plot(lsgk2(t), t=180..Osk,0..100,color=[brown]):
  K5:=plot(lsgk(t), t=0..180,0..100,color=[brown]):
  M4:=plot(lsgm2(t), t=180..Osk,0..100,color=[orange]):
  M5:=plot(lsgm(t), t=0..180,0..100,color=[orange]):
  macro(milchkaffee = COLOR(RGB, 1, 0.8, 0.4)):
  MI4:=plot((lsgm2(t)+9*lsgk2(t))/10,
  t=Osk..600,0..100,color=milchkaffee):
  R1:=plot(20,t=0..180,color=wheat):
  R2:=plot(10,t=180..600,color=wheat):
  macro(milchkaffeeol = COLOR(RGB, 0.7, 0.4, 0.2)):
  OL4:=plot(lsgolga(t),t=180..600,0..100,color=milchkaffeeol):
  OL5:=plot(lsgol(t),t=0..180,0..100,color=milchkaffeeol):
  display({K4,K5,M4,M5,MI4,R1,R2,OL4,OL5},title=`Vergleich der
  Verfahren (9:1) (2)`);
```



Hier sieht man leider fast keinen Unterschied, daher das Verfahren im Ausschnitt.

```
> TE:=plot(42,t=180..600,color=red):
  display({K4,M4,MI4,TE,OL4},view=[420...460,40..50],axes=boxed,ti
  tle=‘Vergleich der Verfahren im Ausschnitt (2)‘);
```



Im Gegensatz zum ersten Beispiel wird die Milch erst drei Minuten lang wärmer, bis auf etwa 11˚C Danach liegt die Umgebung bei nur noch 10 ˚C, und daher kühlt die Milch wieder ab. Dies ist der Grund, dass Oskars Methode hier schneller ist.
Daher zeichnen wir jetzt nocheinmal die Milchkurve:

```
> display({M4,M5,R2},view=[160...200,10.5..11.5],axes=boxed,title=
  ‘Milchkurve im Ausschnitt‘);
```


Milchkurve im Ausschnitt
11.4
11.2
11
10.8
10.6
160
170
180
190
200
t

>